

ЛАБОРАТОРНА РОБОТА 5

Приклади економіко-математичних моделей

1. Приклади типових економіко-математичних задач.
 - 1.1. Задача про оптимальний розподіл капіталу.
 - 1.2. Коефіцієнт Джині та його обчислення.
2. Виробнича функція з прикладами застосування.
 - 2.1. Виробнича функція.
 - 2.2. Задача оптимального розподілу часу на роботу та дозвілля.

1. Приклади типових економіко-математичних задач

1.1. Задача про оптимальний розподіл капіталу

Є вільний капітал розміром 1 млрд. грошових одиниць, який можна використати двома шляхами:

- розмістити у банку як депозит під 20 % річних;
- інвестувати у виробництво з ефективністю 40 %, причому виробничі затрати очікуються пропорційними квадрату обсягу інвестиції, при цьому прибуток оподатковується в розмірі p %.

Питання полягає в тому, за яких значень оподаткування p частину капіталу стає вигідно інвестувати у виробництво. Принагідно важливо також встановити, яку саме частину капіталу слід інвестувати у виробництво, а яку покласти в банк під відсотки?

> **restart:**

Припустимо, що $x \leq 1$ – доля стартового капіталу, яка інвестуватиметься, тоді решта $1-x$ кладеться у банк під 20 % річних.

Цей депозит через рік складатиме такий капітал:

> **dep:=(1-x)*1.2;**

$$dep := 1.2 - 1.2 x \quad (5.1)$$

Тоді як капітал, вкладений у виробництво з урахуванням виробничих витрат та оподаткування через рік складатиме:

> **man:=(1.4*x-a*x^2)*(1-p/100);**

$$man := (1.4 x - a x^2) \left(1 - \frac{1}{100} p \right) \quad (5.2)$$

де a – деякий коефіцієнт пропорційності для виробничих витрат, який лежить у межах $0 < a < 0.4$, причому за більших значень цього коефіцієнта інвестиція вочевидь неефективна, бо виробничі витрати перевищують чистий прибуток.

Загальний капітал через рік складатиме:

> **cap:=collect(dep+man, x);**

$$cap := 1.2 - 1. a (1. - 0.01000000000 p) x^2 + (0.2 - 0.01400000000 p) x \quad (5.3)$$

Відшукаємо значення x , яке забезпечує екстремальне значення виразу (5.3) за умови, що $0 \leq x \leq 1$.

> **eq:=diff(cap,x)=0; x_m:=solve(eq, x);**

$$eq := -2. a (1. - 0.01000000000 p) x + 0.2 - 0.01400000000 p = 0 \quad (5.4)$$

$$x_m := \frac{0.1000000000 (-100. + 7. p)}{a (-100. + p)}$$

Оскільки це повинен бути максимум, друга похідна повинна бути від'ємною:

> **diff(cap, x\$2)<0;**

$$-2. a (1. - 0.01000000000 p) < 0 \quad (5.5)$$

що забезпечене при будь-якому позитивному значенні коефіцієнту $a > 0$.

> $x_m := \text{unapply}(x_m, p, a);$

$$x_m := (p, a) \rightarrow \frac{0.1000000000 (-100. + \sqrt{a (-100. + p)})}{a (-100. + p)} \quad (5.6)$$

Зауважимо, що незалежно від значення a вираз (5.6) має корінь для значення:

> $p_{cr} := \text{evalf}(100/7);$

$$p_{cr} := 14.28571429 \quad (5.7)$$

Якщо ставка оподаткування перевищує цю критичну межу, то інвестиції у виробництво стають економічно не вигідними, формально при цьому $x_m < 0$, і слід весь капітал класти в банк під відсотки. За меншої, ніж критична, ставки оподаткування певну частину капіталу слід інвестувати у виробництво.

Побудуємо послідовність значень величини x_m з виразу (5.6) для різних значень коефіцієнту виробничих витрат a :

> $X := \text{seq}(x_m(p, 0.05 \cdot n), n = 1..8);$

Далі створимо графіки побудованої послідовності, обмежившись лише тими значеннями її факторів, які потрапляють в інтервал $0 \leq x_m \leq 1$:

> $\text{plot}([X, 1, 0], p = 0..20, \text{view} = -0.2..1, \text{axes} = \text{boxed}, \text{thickness} = 2, \text{gridlines} = \text{true}, \text{font} = [\text{Arial}, 14], \text{labels} = [p, x_m], \text{labelfont} = [\text{Arial}, 14], \text{labeldirections} = [\text{horizontal}, \text{vertical}], \text{tickmarks} = [10, 7], \text{caption} = \text{Рис. 1. Залежність долі капіталу, що інвестується \n у виробництво від оподаткування при різних \n коефіцієнтах витрат a=0.05..0.4});$




Рис. 5.1. Залежність долі капіталу, що інвестується у виробництво, від оподаткування при різних коефіцієнтах витрат $a = 0.05..0.4$

З графіку ясно видно, що всі криві на рис. 5.1 перетинаються в точці, яка є їх спільним нулем. Іншими словами всі члени послідовності мають спільний корінь, який визначений виразом (5.7). З графіків також видно, що чим меншими є виробничі витрати (коефіцієнт a), тим більшу долю капіталу варто вкладати у виробництво. Зокрема за умови ставки оподаткування нижче 7 % та норми виробничих витрат < 0.055 можна інвестувати навіть весь стартовий капітал. Вираз (5.7) дозволяє розрахувати оптимальну долю інвестиційного капіталу за умов різного оподаткування p та норм виробничих витрат a . Зокрема для такого конкретного випадку як $p = 8$, $a = 0.1225$ маємо:

> $x_m(9, 0.12);$

$$0.3388278388. \quad (5.8)$$

Тобто за таких умов інвестувати у виробництво найвигідніше приблизно третину стартового капіталу.

1.2. Коефіцієнт Джині та його обчислення

Коефіцієнт (індекс) Джині – показник нерівності розподілу серед населення деякої величини, наприклад, річних доходів. Коефіцієнт Джині може мати значення в інтервалі між 0 і 1, або ж від 0 до 100 %, причому 0 означає абсолютну рівність – розподілена величина приймає лише одне значення для всіх, а 1 позначає повну нерівність розподілу величини.

Найбільш відомим коефіцієнт Джині є як економічна міра нерівності доходів домогосподарств деякої країни чи регіону. Коефіцієнт Джині для доходів домогосподарств є найпопулярнішим показником економічної нерівності в кожній країні (рис. 5.2).




Рис. 5.2. Коефіцієнт Джині розподілу доходів для країн світу, згідно з даними 2009 року [22]

Коефіцієнт Джині найпростіше визначити за допомогою так званої кривої Лоренца, що зображує частку величини y %, яка зосереджується на x % популяції з найменшим значенням цієї величини (рис. 5.3). Наприклад, для розподілу доходів точка (30 %, 10 %) буде лежати на кривій Лоренца, якщо сукупний дохід 30 % найбідніших сімей рівний десяти процентам сукупного доходу усіх домогосподарств.




Рис. 5.3. Крива Лоренца (OABC)

Коефіцієнт Джині рівний відношенню площі області утвореної кривою Лоренца і прямою повної рівності (прямою під кутом 45°), тобто площі фігури OABC до площі трикутника ODC, утвореного прямою повної рівності і прямими $y=0$ та $x=1$. На рис. 5.3 перша область позначена сірою штриховкою. Якщо позначити площі відповідних фігур A і B, то можна записати формулу для коефіцієнта Джині у вигляді:

$$G = \frac{A}{A+B}. \quad (5.9)$$

Оскільки площа трикутника $A + B = \frac{1}{2}$, то отримаємо такі формули:

$$G = 2A = 1 - 2B. \quad (5.10)$$

Якщо весь дохід є рівномірно розподілений, то крива Лоренца збігається з прямою повної рівності і значення коефіцієнта Джині дорівнює нулю. Чим більшим є значення коефіцієнта Джині, тим нерівномірніше дохід розподілений серед домогосподарств певної країни.

Розглянемо деякий приклад. Припустимо, що для певної країни крива Лоренца може бути описана таким рівнянням:

> restart: y:=1-(1-x)^(0.5);

$$y := 1 - (1 - x)^{0.5} \quad (5.11)$$

Зобразимо криву Лоренца та пряму повної рівності, виділивши площу під кривою Лоренца (фігура B) кольором:

> plot([y,x], x=0..1, view=0..1, color=[khaki,navy], gridlines=true, axes=boxed, font=[Arial,14], labels=["x", "y(x)", "B"], labelfont=[Arial,14], labeldirections=[horizontal, vertical], filled=[true,false], thickness=[2,2], caption="Рис. 4. Крива Лоренца \n та площа під нею (B)");




Рис. 5.4. Крива Лоренца та площа під нею (B)

Знаходимо площу під кривою Лоренца та коефіцієнт Джині:

> Int(y,x=0..1)=int(y,x=0..1); B:=rhs(%);

$$\left(\int_0^1 (1 - (1 - x)^{0.5}) dx = 0.3333333333 \right) \quad (5.12)$$

$$B := 0.3333333333$$

> G:=1-2*B;

$$G:=0.3333333334. \quad (5.13)$$

Коефіцієнт Джині (5.13) приблизно відповідає ситуації в Україні, як це слідує з рис. 5.2. Для порівняння, коефіцієнт Джині для Китаю у 2012 році складав 0.474, в Росії – 0.422, в США – 0.477.

Доволі часто характеристикою нерівності розподілу доходів серед населення слугує також відношення сукупних доходів 20 % найбагатшої частини населення до найбідніших 20 %. Визначимо цей показник.

> a:=int(y,x=0.8..1)/int(y,x=0..1); # Сукупна доля 20% найбагатшого населення

> b:=int(y,x=0..0.2)/int(y,x=0..1); # Сукупна доля 20% найбіднішої верстви населення

g:=a/b; # Показник нерівності

$$a := 0.4211145618,$$

$$b := 0.03108350559,$$

$$g := 13.54784648.$$

Отже, найбагатшим 20 % суспільства належить приблизно 42 % сукупного річного доходу, тоді як найбіднішим 20 % населення – лише трохи більше 3 % такого доходу. Для розглянутої в нашому прикладі умовної країни відносний показник соціальної нерівності між цими соціальними групами є вищим від такого $g \approx 7$, який вважається граничним для виникнення надмірної соціальної напруги в суспільстві.

2. Виробнича функція з прикладами застосування

2.1. Виробнича функція

Виробнича функція (ВФ) – залежність між кількістю використаних у виробництві ресурсів (факторів виробництва) та обсягом виробленої продукції (або послуг). Виробничі функції застосовують для аналізу впливу різних факторів на обсяг випуску продукції на мікро- та макрорівнях економіки – від окремої фірми до народного господарства країни в цілому, тоді значенням ВФ є валовий національний продукт (ВВП) або валовий національний дохід (ВНД).

У якості ресурсів, тобто змінних ВФ, найчастіше використовують накопичений труд у формі виробничих фондів K -капіталу та справжній живий труд L , а економіка на макрорівні моделюється двофакторною нелінійною виробничою функцією X :

$$X = F(K, L), \quad (5.14)$$

яка стверджує, що випуск продукції є функцією від витрат ресурсів: фондів (капіталів) та праці. Замість загальної формули (5.14) в економічних аналізах найчастіше використовують два часткових випадки ВФ:

1. Мультиплікативна форма (МВФ) функції:

$$X = AK^\alpha L^\beta \quad 0 < (\alpha, \beta) < 1; \quad (5.15)$$

2. ВФ Кобба-Дугласа:

$$X = AK^\alpha L^{1-\alpha} \quad 0 < \alpha < 1. \quad (5.16)$$

При цьому величину $\frac{X}{L}$ називають середньою ефективністю праці, а $\frac{X}{K}$ – середньою віддачею фондів.

ВФ є неокласичною, якщо вона задана для всіх невід'ємних значень аргументів (K, L) , є неперервною функцією аргументів, яка має похідні потрібного порядку і відповідає трьом економічним вимогам:

$$F(0, L) = F(K, 0) = 0, \quad (5.17)$$

що означає неможливість виробництва без одного з ресурсів. Окрім того:

$$\frac{\partial F}{\partial K} > 0, \quad \frac{\partial F}{\partial L} > 0, \quad (5.18)$$

що означає вимогу зростання виробництва за умови зростання ресурсів.

$$\frac{\partial^2 F}{\partial K^2} < 0, \quad \frac{\partial^2 F}{\partial L^2} < 0, \quad (5.19)$$

що означає уповільнення зростання виробництва за умов збільшення лише одного ресурсу за фіксованої кількості іншого. Очевидно, що ВФ типу (5.15), (5.16) є неокласичними.

2.2. Задача оптимального розподілу часу на роботу та дозвілля

2.2.1. Розподіл без урахування оподаткування

Припустимо, що повна кількість часу, якою ми диспонуємо, позначена через T , а кількість часу, яка присвячена дозвіллю – L . Тоді кількість часу, відведена на роботу, становить $h = T - L$.

Заробітна плата за годину позначається w . Зауважимо, якщо людина працює h годин при заробітній платі за годину w , тоді дохід складатиме:

$$wh = w(T - L) = wT - wL.$$

Ми також вважаємо, що дохід від праці витрачається на споживання, зокрема й на

дозвілля, тобто $wT - wL = P_C \cdot C$, де P_C – ціна споживання, а C – обсяг споживання. Покладемо для простоти ціну споживання одиниці $P_C = 1$, отже:

> restart; w*T-w*L-C=0;

$$-L w + T w - C = 0 \quad (5.20)$$

що просто означає, що бюджет повинен виконуватися і ніхто не може споживати більше, ніж зароблено. Тепер припустимо, що споживач має корисність від обох ресурсів: як від роботи, так і від дозвілля, і ця корисність задається виробничою функцією типа Кобба-Дугласа:

> U(C,L)=C^beta*L^(1-beta);

$$U(C, L) = C^\beta L^{1-\beta} \quad (5.21)$$

яка може бути зображена графічно:

> plot3d(eval(C^beta*L^(1-beta), beta=0.5), C=0..800, L=0..24, axes=framed, style=patchcontour, color='Niagara Azure', contours = 12, font=[Arial,14], orientation=[-30,45,0], labels=["Споживання", "Дозвілля", "Корисність"], labelfont=[Arial,14], labeldirections=[horizontal, horizontal, vertical], title='Рис. 5. Функція корисності споживача \n від споживання та дозвілля');




Рис. 5.5. Функція корисності споживача від споживання та дозвілля

Заважимо, що уздовж контурів на рис. 5.5 значення функції корисності є незмінними.

Поставимо таку задачу: максимізувати функцію корисності споживача шляхом оптимального розподілу часу поміж роботою та дозвіллям. Задачу на умовний екстремум розв'яжемо спочатку відомим з математики методом множників Лагранжа:

> restart;

con:=-L*w+T*w-C; # Умови

ob:=C^beta*L^(1-beta); # Цільова функція корисності

LL:=con*lambda+ob; # Лагранжіан

$$con := -L w + T w - C \quad (5.22)$$

$$ob := C^\beta L^{1-\beta}$$

$$LL := (-L w + T w - C) \lambda + C^\beta L^{1-\beta}$$

Умовами екстремуму лагранжіана є такі рівняння:

> eq1:=diff(LL,C)=0; eq2:=diff(LL,L)=0;

$$eq1 := -\lambda + \frac{C^\beta \beta L^{1-\beta}}{C} = 0 \quad (5.23)$$

$$eq2 := -w \lambda + \frac{C^\beta L^{1-\beta} (1-\beta)}{L} = 0$$

З відповідними розв'язками рівнянь (5.23) для множника Лагранжа λ :

> solve(eq1,lambda); solve(eq2,lambda);

$$\frac{C^\beta \beta L^{1-\beta}}{C} \quad (5.24)$$

$$- \frac{C^\beta L^{1-\beta} (-1 + \beta)}{L w}$$

які після прирівнювання формують рівняння:

> eq3:=%=%%;

$$eq3 := - \frac{C^\beta L^{1-\beta} (-1 + \beta)}{L w} = \frac{C^\beta \beta L^{1-\beta}}{C} \quad (5.25)$$

> C_1:=solve(eq3,C);

$$C_1 := - \frac{\beta L w}{-1 + \beta} \quad (5.26)$$

Підставимо вираз (5.26) в умови нульового бюджету (5.22) поділеного на годинну заробітну платню w :

> eq4:=simplify(eval(con/w=0, [C=C_1]));

$$eq4 := \frac{T \beta + L - T}{-1 + \beta} = 0 \quad (5.27)$$

> L_opt:=solve(eq4, L); # Вираз для оптимальної кількості дозвілля

$$L_{opt} := -T (-1 + \beta) \quad (5.28)$$

> C_opt:=eval(C_1, [L=L_opt]);

$$C_{opt} := \beta T w \quad (5.29)$$

Методами Maple та ж задача розв'язується числовими алгоритмами, що демонструє код, наведений нижче.

> restart; with(Optimization);

[ImportMPS, Interactive, LPSolve, LSSolve, Maximize, Minimize,
NLPSolve, QPSolve]

> beta:=.5; w:=50; T:=24; # Числові параметри задачі (показник цільової функції, погодинна оплата праці та кількість годин на добу)

$$\beta := 0.5 \quad (5.30)$$

$$w := 50$$

$$T := 24$$

> U:=C^beta*L^(1-beta); # Цільова функція корисності

con:=-L*w+T*w-C; # Умова

$$U := C^{0.5} L^{0.5} \quad (5.31)$$

$$con := -C - 50 L + 1200$$

> Opt1:=evalf(Maximize(U, [con = 0]), 4);

Максимізація цільової функції за умови

$$Opt1 := [84.85, [C = 600.0, L = 12.]] \quad (5.32)$$

З останньої операції можна побачити, що максимум функції корисності споживача на рівні $U = 84.85$ забезпечує споживання розміром 600 грошових одиниць та дозвілля впродовж 12 годин на добу. Ті самі показники можна отримати і за формулами (5.28)-(5.29).

```
> C=evalf(w*T*beta, 4);
L=evalf((1-beta)*T, 3); # Розрахунки оптимального споживання та дозвілля по формулах (5.28) та (5.29)
```

$$C = 600.0 \quad (5.33)$$

$$L = 12.0$$

Зобразимо результати графічно:

```
> `Оптимальний обсяг споживання`=evalf(w*T*beta, 2);
`Оптимальний обсяг дозвілля`=evalf((1-beta)*T, 2);
v1:=plot(-L*w+T*w, L=0..35, color=blue, thickness=3): v2:=plots[contourplot](C^beta*L^(1-
beta), L=0..35, C=0 ..1400, contours=60, color=khaki):
v3:=plots[textplot]([12, 600, "Лінія доходу та оптимальна позиція"], align={ABOVE,
RIGHT}, font=[Helvetica, bold, 12]):
v4:=plots[pointplot]([12, 600], axes=boxed, symbol=solidcircle, symbolsize=20):
plots[display]({v1, v2, v3, v4}, labels=["Дозвілля", "Споживання"], labelfont=[Arial,14],
labeldirections=[horizontal, vertical], font=[Helvetica, 14], view=[5..25, 50..1200],
tickmarks=[12,7], gridlines=true);
```

*Оптимальний обсяг споживання = 600.
Оптимальний обсяг дозвілля = 12.*




Рис. 5.6. Споживання та дозвілля, їх оптимізація

На рис. 5.6 контури показують різні значення цільової функції корисності, точкою на лінії доходів показано оптимальний розподіл між роботою (12 год.) та дозвіллям (12 год.), який забезпечує обсяг споживання у 600 грошових одиниць на добу, з максимального значення функції корисності $U = 84.85$.

2.2.2. Вплив податків на споживання та розподіл часу

Тепер врахуємо оподаткування. Припустимо, що доля податків складає t з кожної грошової одиниці, іншими словами, погодинна оплата з урахуванням податків складає тепер $w(1-t)$ грошових одиниць. Бюджетна умова з урахуванням податків виглядатиме так:

$$(1-t)wT - (1-t)wL = C. \quad (5.34)$$

Застосуємо програму оптимізації, поклавши долю податків у 40 %.

```
> restart:
with(Optimization); with(plots):
```


[*ImportMPS, Interactive, LPSolve, LSSolve, Maximize, Minimize, NLPSolve, QPSolve*]

```
> beta:=.5;
t:=.4;
w:=50;
T:=24;
U:=C^beta*L^(1-beta);
con:=(1-t)*w*T-C-(1-t)*w*L;
```

$$\beta := 0.5 \quad (5.35)$$

$$t := 0.4$$

$$w := 50$$

$$T := 24$$

$$U := C^{0.5} L^{0.5}$$

$$con := 720.0 - C - 30.0 L$$

```
> Opt2:=evalf(Maximize(U, [con = 0]), 2);
```

$$Opt2 := [66., [C = 360., L = 12.]] \quad (5.36)$$

Як видно з (5.36), податки зменшують як обсяг споживання так і максимальне значення функції корисності для споживача $U_2 = 66 < U_1 = 85$. Відобразимо це порівняння у вигляді графіків:

```
> "Оптимальне споживання з податками"=360; "Оптимальне дозвілля з
податками"=12;
v1:=plot(-L*w+T*w, L=0..35, color=navy, thickness=3): v2:=contourplot(C^beta*L^(1-beta),
L=0..35, C=0..1400, contours=60, color=khaki):
v3:=textplot([12, 600, "Оптимальна позиція без податків"], align={ABOVE, RIGHT},
font=[Helvetica, bold, 12]):
v4:=pointplot({[12, 600]}, axes=boxed, symbol=solidcircle, symbolsize=20, color=blue):
v5:=plot((1-t)*w*T-(1-t)*w*L, L=0..35, color=`Niagara Burgundy`, thickness=3):
v6:=textplot([12, 360, "Оптимальна позиція з податками"], align={ABOVE, RIGHT},
font=[Helvetica, bold, 12]):
v7:=pointplot({[12, 360]}, axes=boxed,
symbol=solidcircle, symbolsize=20, color=red): display({v1, v2, v3, v4, v5, v6, v7},
labels=["Дозвілля", "Споживання"], labelfont=[Arial,14],
labeldirections=[horizontal,vertical], font=[Helvetica, 14], view=[5..25, 50..1200],
tickmarks=[12,7], gridlines=true);
```

"Оптимальне споживання з податками" = 360

"Оптимальне дозвілля з податками" = 12




Рис. 5.7. Споживання та дозвілля з урахуванням податків, їх оптимізація

Завдання для самостійного виконання (виконати варіанти задач відповідно до Вашого номера у списку групи).

I. Задача про оптимальний розподіл капіталу

Є вільний капітал розміром 1 млрд. грошових одиниць, який можна використати двома шляхами:

- розмістити у банку як депозит під α % річних;
- інвестувати у виробництво з ефективністю β %, причому виробничі затрати очікуються пропорційними квадрату обсягу інвестиції, при цьому прибуток оподатковується в розмірі p %.

Питання полягає в тому, за яких значень оподаткування p частину капіталу стає вигідно інвестувати у виробництво. Принагідно важливо також встановити, яку саме частину капіталу слід інвестувати у виробництво, а яку покласти в банк під відсотки?

1. $\alpha = 15\%$, $\beta = 35\%$.
2. $\alpha = 16\%$, $\beta = 44\%$.
3. $\alpha = 18\%$, $\beta = 37\%$.
4. $\alpha = 21\%$, $\beta = 42\%$.
5. $\alpha = 22\%$, $\beta = 39\%$.
6. $\alpha = 23\%$, $\beta = 41\%$.
7. $\alpha = 24\%$, $\beta = 43\%$.
8. $\alpha = 25\%$, $\beta = 45\%$.

II. Задача оптимального розподілу часу на роботу і дозвілля

Розподіл без урахування оподаткування

1. $\beta = 0,45$; $w = 55$; $T = 24$.

2. $\beta = 0,55; w = 60; T = 24.$

3. $\beta = 0,60; w = 70; T = 24.$

4. $\beta = 0,65; w = 80; T = 24.$

Розподіл з урахуванням оподаткування

5. $\beta = 0,45; w = 75; t = 35\%; T = 24.$

6. $\beta = 0,55; w = 80; t = 45\%; T = 24.$

7. $\beta = 0,60; w = 90; t = 38\%; T = 24.$

8. $\beta = 0,65; w = 100; t = 42\%; T = 24.$